
Тема «Информационная безопасность», изучаемая 
в школе, предполагает знакомство учащихся с неко-
торыми алгоритмами шифрования. Рассматриваются 
преимущественно ручные способы шифрования и де-
шифрования наиболее простых алгоритмов: «Скитала», 
шифр Цезаря, шифр Виженера.

Не всегда у учителя есть возможность запрограмми-
ровать хотя бы один из таких алгоритмов, поэтому в ос-
новном знакомство учащихся основывается на ручном 
преобразовании. К сожалению, не всегда у ребят хватает 
терпения и внимательности для полной процедуры 
шифрации и дешифрации некоторого осмысленного 
сообщения. В этой связи хочется предложить в каче-
стве тренировки процессор электронных таблиц как 
наиболее доступное средство в школе для составления 
алгоритмов. Есть много способов реализации крипто-
графических алгоритмов с использованием различных 
сред программирования (C++, Pascal, Mathcad и др.), 
например, описанных в [1, 2], но электронные табли-
цы — это один из самых простых и наглядных средств 
для решения таких задач.

Задача 1. Алгоритм «Скитала».

Рассмотрим алгоритм «Скитала», являющийся од-
ним из древнейших способов шифрования сообщений. 
Его придумали и использовали в Древней Спарте. Суть 
метода состояла в написании букв в определенной по-
следовательности, которую можно было расшифровать, 

намотав  узкую полоску пергамента, на которой писалось 
сообщение, на специальную палочку. При этом буквы 
выстраиваются в ряд в зависимости от толщины палочки.

Процесс шифрования текста заключался в переста-
новке букв в соответствии с длиной окружности палочки. 
Палочка и была ключом к разгадке шифра. Условно 
длина окружности палочки — это количество символов 
в строке шифруемого текста, а длина палочки — это 
число столбцов в таблице (рис. 1).

Рис. 1. Матрица с соотношениями окружности и длины 
палочки-скиталы

В данном случае на палочке размещаются четыре 
буквы и при кодировании сообщения «нас атакуют*» 
получается «науатюсат к*».

В ячейку B3 вводится фраза для шифрования.
В ячейку C4 вводится количество букв, которые об-

разуют полный виток палочки (n = 4).
Далее строится матрица размером 10 × 10 (полагаем, 

что палочки большего размера трудно было утаить от 

противника). Строки и столбцы матрицы последова-
тельно нумеруются.

Ячейки С6:L6 и B7:B16 отведены для нумерации 
соответственно столбцов и строк матрицы шифро-
вания.

В ячейку С7 записывается формула:

=ЕСЛИ(C$6<=$C$4;ПСТР($B$3;($B7-1)*$C$4+ 
C$6;1);"")

Рассмотрим данную формулу более подробно. В свя-
зи с тем что формула будет скопирована на множество 
ячеек как по горизонтали, так и по вертикали, исполь-
зуются абсолютная и смешанная адресации ячеек. Если 
номер столбца ячейки (C$6) меньше или равен числу 
n ($C$4), то из шифруемого сообщения ($B$3) берем 
символ с позиции, равной номеру текущей строки без 
единицы, умноженному на n, плюс номер текущего 
столбца. Если же номер столбца ячейки превышает число 
n, то просто ставим в этой ячейке пустую строку «». Для 
поиска отдельного символа, стоящего на заданной по-
зиции, использована функция:

ПСТР(текст; начальная_позиция;число_знаков)

Далее формула легко копируется по строкам и столб-
цам.

Осталось собрать все буквы по столбцам (ячейка 
C17). Сделать это можно с помощью оператора слияния 
строк (конкатенации) &. В нашем случае для первого 
столбца формула выглядит так:

=C7&C8&C9&C10&C11&C12&C13&C14&C15&C16

После этого формула из ячейки С17 копируется 
в ячейки D17:L17.

Зашифрованное сообщение (С20) образуется путем 
слияния строк ячеек C17:L17, формула:

=C17&D17&E17&F17&G17&H17&I17&J17&K17&L17

Общая схема реализации алгоритма в электронных 
таблицах выглядит, как представлено на рисунке 2.

Дешифровка сообщения осуществляется путем 
подбора параметра m, являющегося предполагаемой 
длиной окружности палочки (количества букв, которые 

выводятся на одной стороне палочки). После падения 
Спарты Аристотель предложил оригинальный способ 
дешифровки сообщений: вместо цилиндра использо-
вать конус. Тогда можно последовательно передвигать 
скиталу, пока не получится осмысленное сообщение.

Однако при реализации в Excel можно просто под-
бирать количество букв в отдельной ячейке.

В ячейку B23 вводится зашифрованный текст. 
Возьмем в качестве примера сообщение «ЭФВПТРНА-
ОДЕРШРЙТИЕСЫ».

В ячейку C24 введем число m — предполагаемый 
ключ. Пусть m = 3. Далее строится матрица 10 × 10, обя-
зательно нужно подписать строки и столбцы (рис. 3).

Расчетная формула в матрице будет следующей:

=ЕСЛИ($B27<=$C$24;ПСТР($B$23;(C$26-1)* 
$C$24+$B27;1);"")

Здесь также проверяется условие, что номер строки 
не должен превышать число m. Если условие выпол-
няется, то в ячейку вводится буква зашифрованного 
сообщения, стоящая на позиции (номер столбца – 1) × 
× (m + номер строки). В случае, если номер строки пре-
вышает m, просто оставляем в ячейке пустую строку. 
Также ее нужно растянуть на все ячейки матрицы по 
горизонтали и вертикали.

При предполагаемом m = 3 мы не получим осмыс-
ленного сообщения. Тогда просто m увеличим на 1, и вот 
уже получилось осмысленное выражение (рис. 3).

Рис. 3. Общая схема дешифрования алгоритма «Скитала»

Рис. 2. Общая схема реализации алгоритма «Скитала»

Задача 2. Шифр Цезаря.

Еще одним знаменитым древним шифром является 
шифр Цезаря. Каждая буква в сообщении заменялась на 
другую букву, которая в данном алфавите отстоит от ис-
ходной на несколько позиций дальше. Например, буква 
А зашифрованного сообщения заменялась со сдвигом 
три на букву Г, буква Б — на букву Д и т. д. Количество 
букв в сдвиге и является ключом к шифру. Такой шифр 
легко взломать. Усложнением алгоритма является про-
извольно выбранное число n — величина сдвига.

Рассмотрим пример реализации алгоритма в Excel.
Пусть требуется зашифровать текст «АПЕЛЬСИН». 

Введем это слово в ячейку B2, в ячейку B3 введем алфа-
вит, в ячейку B4 — ключ смещения.

Построим матрицу, в которой возьмем нумерацию 
строк от 1 до 33 (количество букв в русском алфавите), 
а нумерацию столбцов возьмем с некоторым запасом, 
чтобы хватало на длинные сообщения, например, от 
1 до 40. Для решения данной задачи учащимся следует 
получить справочную информацию по дополнительным 
двум функциям работы с текстовыми строками:

ДЛСТР(текст) —возвращает количество знаков 
в текстовой строке;

ПОИСК(искомый_текст;текст_для_поиска;нач_по-
зиция) — находит одну текстовую строку внутри второй 
текстовой строки и возвращает номер начальной по-
зиции первой строки, отсчитывая его от первого знака 
второй строки [4].

Формула, вводимая в ячейку B6, будет несколько 
громоздкой, поэтому рассмотрим ее по частям.

1) Получение отдельного символа из исходного 
сообщения ($B$2), соответствующего номеру 
текущего столбца:

ПСТР($B$2;B$5;1)

2) Поиск позиции символа, полученного в резуль-
тате действия 1 в алфавите. При этом учтено, что 
при символах, стоящих ближе к концу алфавита, 
должен происходить циклический сдвиг. Для этого 
достаточно продублировать строку с алфавитом 
($B$3&$B$3):

ПОИСК(ПСТР($B$2;B$5;1);$B$3&$B$3)

3) К позиции, полученной в результате действия 2, 
добавим ключ смещения:

ПОИСК(ПСТР($B$2;B$5;1);$B$3&$B$3)+$A6

4) Найдем символ в алфавите, стоящий на позиции, 
полученной в действии 3. При этом также проду-
блируем исходную строку с алфавитом на случай 
возможных сдвигов:

ПСТР($B$3&$B$3;ПОИСК(ПСТР($B$2;B$5;1); 
$B$3&$B$3)+$A6;1)

5) Запишем общую формулу, которая предполагает, 
что действие 4 выполняется, если номер текущего 
столбца меньше либо равен длине исходного со-
общения, в противном случае выводим в ячейку 
пустую строку:

=ЕСЛИ(B$5<=ДЛСТР($B$2);ПСТР($B$3&$B
$3;ПОИСК(ПСТР($B$2;B$5;1);$B$3&$B$3)+
$A6;1);"")

Рис. 4. Общая схема алгоритма шифрования Цезаря

Для удобства нахождения зашифрованной строки 
можно воспользоваться условным форматированием 
в Excel:

•	 выделить номера строк (A6:A38);
•	 Условное форматирование;
•	 Правила выделения ячеек;
•	 Равно;
•	 $B$4.
Результат показан на рисунке 4.
Для дешифрации алгоритма в ячейку B41 вводит-

ся зашифрованный текст. Для примера взята строка 
ЧЖЗЁЪМХЖДЖЗДЮЗЖФГХЦЪЯАЪЁЖЗЁЮЗ. Также 
строится матрица: нумеруются строки от 1 до 33, столбцы 
от 1 до 40 с учетом длины передаваемого сообщения. 
Расчетная формула будет изменена. Во-первых, следует 
учесть, что при дешифрации из позиции, полученной 
в действии 2, нужно вычесть ключ. Во-вторых, при вычи-
тании ключа мы можем легко уйти за пределы алфавита 
(пример с буквой А), поэтому искать нужную позицию 
будем относительно середины двойного алфавита (по-
зиция 34):

ПОИСК(ПСТР($B$41;B$5;1);$B$3&$B$3;34)

Результирующая формула будет следующей: 

=ЕСЛИ(B$43<=ДЛСТР($B$41);ПСТР($B$3&$B
$3;ПОИСК(ПСТР($B$41;B$5;1);$B$3&$B$3;34)-
$A44;1);"")

Ее нужно скопировать на всю матрицу. Также мож-
но воспользоваться условным форматированием для 
выделения номера строки, соответствующего ключу. 
В этой строке и будет наше расшифрованное сообще-
ние (рис. 5).

Задача 3. Шифр Виженера.

Шифр Виженера был известен с XVI века и почти 
300 лет его не могли разгадать. Шифр использует по сути 
несколько шифров Цезаря за счет того, что позиции 
смещения определяются не по одному числу, а по набору 
чисел, соответствующих некоторому ключевому слову.

Например, пусть ключевое слово «ЯБЛОКО». Вы-
пишем позиции каждой буквы этого слова из русского 
алфавита: 33, 2, 13, 16, 12, 16. Эти числа определяют 
смещение каждого символа в шифруемом сообщении.

Пусть зашифровать нужно слово «КРИПТОСТОЙ-
КОСТЬ». Тогда в соответствии с позициями смещения 
получится: К → Й (33 позиции), Р → С (2 позиции), 
И → Ф (13 позиций), П → Ю (16 позиций), Т → Э (12 по-
зиций), О → Э (16 позиций) и т. д. по кругу.

Данный шифр не сохраняет частоту, а следовательно, 
менее подвержен частотному анализу, с помощью кото-
рого часто дешифруются простые шифры. Более того, 
сам Виженер подчеркивал, что нужно брать в качестве 
ключа не осмысленное слово, а случайный набор букв.

Для шифрования используется tabula recta (квадрат 
Виженера). Несмотря на то что при реализации в Excel 
можно встретить решение без использования квадрата 
Виженера, вычисляющего новый символ путем добавле-
ния позиции символа из таблицы ASCII, в нашем случае 
будет использован классический алгоритм. Дело в том, что 
в алфавите могут быть использованы помимо букв цифры 
и другие символы, что обусловливает более качественное 
и универсальное шифрование. В некоторых случаях вместо 
алфавитной последовательности символов можно исполь-
зовать другие последовательности, например, в квадрате 
Бофора использовался алфавит, написанный наоборот.

Рис. 5. Общая схема алгоритма дешифрования Цезаря

При получении зашифрованного символа ci исполь-
зуется формула:

сi = (mi + ki) mod n,

где mi — позиция исходного символа в алфавите, ki — 
позиция символа ключа в алфавите, n – мощность 
алфавита.

При дешифровке сообщения используется формула:

mi = (ci + n – ki) mod n.

Введем в ячейки следующие данные:
•	 B1 — шифруемое сообщение, например, «КРИП-

ТОСТОЙКОСТЬ»;
•	 B2 — ключ, например, «ЯБЛОКО»;
•	 B3 — алфавит (можно использовать свой алфавит, 

перенося знаки препинания в начало или конец 
алфавита);

•	 B4 — мощность алфавита, т. е. длина строки ячей-
ки B3: =ДЛСТР(B3).

В столбце C введем номера по порядку, достаточные 
для не очень длинного сообщения (например, длина со-
общения не превосходит 40). Тогда введем нумерацию 
от 1 до 40 в ячейках C1:C40.

Столбец D будет содержать передаваемое сообще-
ние, разбитое посимвольно. Для этого в ячейку D1 
введем формулу:

=ПСТР($B$1;C1;1)

Данную формулу нужно растянуть на ячейки D2:D40.
В столбце E будут находиться позиции каждого 

отдельного символа исходного сообщения в алфави-
те. Во избежание ошибок сделаем это через функцию 
ЕСЛИ, которая проверяет, существует ли в принципе 
очередной символ передаваемого сообщения:

=ЕСЛИ(D1<>"";ПОИСК(D1;$B$3);"").

В столбце F будут находиться символы ключа, по-
вторенные столько раз, сколько нужно для сообщения 
до 40 символов (для повторения ключа использовалась 
операция конкатенации &):

=ЕСЛИ(D1<>"";ПСТР($B$2&$B$2&$B$2&$B$2&
$B$2&$B$2;C1;1);"")

В столбце G находятся позиции символа ключа 
в алфавите:

=ЕСЛИ(F1<>"";ПОИСК(F1;$B$3);"")

В столбце H находятся символы зашифрованного 
текста (так как нужно обойти проблему символа, сто-
ящего на нулевой позиции, берем удвоенный алфавит 
$B$3&$B$3)

=ЕСЛИ(D1<>"";ПСТР($B$3&$B$3;$B$4+ОСТАТ 
(E1+G1;$B$4)-1;1);"")

Осталось собрать полученные зашифрованные 
символы в одну строку с использованием все той же 
конкатенации & ячеек H1:H40.

Рассмотрим дешифровку сообщения. Для каждого 
зашифрованного символа в столбце I вычислим его по-
зицию в алфавите:

=ЕСЛИ(G1<>"";ПОИСК(H1;$B$3);"")

В столбце J запишем дешифрованный символ, полу-
ченный согласно формуле (2):

=ЕСЛИ(H1<>"";ПСТР($B$3&$B$3;ОСТАТ(I1+
$B$4-G1+1;$B$4)+$B$4;1);"")

Общий вид алгоритма показан на рисунке 6.
Таким образом, работа по созданию этих известных 

алгоритмов позволит учащимся понять особенности 
криптографии, легко воссоздать алгоритмы в макси-
мально доступном средстве — среде электронных таблиц, 
познакомиться с некоторыми новыми для себя функци-
ями Excel (ДЛСТР, ПОИСК, ПСТР, &). Данные задачи 
могут быть рассмотрены как на уроке информатики при 
углубленном обучении по теме «Информационная без-
опасность. Шифрование» в X—XI классах [3], так и во 
внеурочной деятельности. На основе такого подхода 
можно организовать и проектную деятельность, заклю-
чающуюся в составлении известных криптографических 
алгоритмов в среде электронных таблиц.  

Рис. 6. Расчетная таблица шифра Виженера в среде электронных таблиц


